
pkgname=guile
pkgver=2.2.0
pkgrel=1
pkgdesc="GNU Ubiquitous Intelligent Language for Extensions, the official extension language for the GNU operating system"
url="https://www.gnu.org/software/guile/"
arch=('x86_64')
license=('GPL')
depends=('gmp' 'libltdl' 'ncurses' 'texinfo' 'gc' 'libffi' 'libunistring')
source=("https://ftp.gnu.org/gnu/guile/${pkgname}-${pkgver}.tar.gz")
options=('!libtool')
md5sums=('0d5de8075b965f9ee5ea04399b60a3f9')

build() {
  cd ${pkgname}-${pkgver}
  
  ./configure --prefix=/usr \
	--disable-static  \
	--disable-error-on-warning
  make 
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR=${pkgdir} install
  
  # stop elfutils warnings
  mkdir -p ${pkgdir}/usr/share/gdb/auto-load/usr/lib
  mv ${pkgdir}/usr/lib/libguile-2.2.so.*gdb.scm ${pkgdir}/usr/share/gdb/auto-load/usr/lib
}
